
Example 6.3. Registered add-and-compare circuit

	--------------------------------------------------------------
library ieee;
use ieee.std_logic_1164.all;
[bookmark: _GoBack]use ieee.numeric_std.all;

entity add_compare_registered is
	generic (
		NUM_BITS: natural := 8);
	port (
		clk: in std_logic;
		a, b: in std_logic_vector(NUM_BITS-1 downto 0);
		comp: out std_logic;
		sum: out std_logic_vector(NUM_BITS downto 0));
end entity add_compare_registered;

architecture rtl of add_compare_registered is
	signal a_uns, b_uns: unsigned(NUM_BITS downto 0);
begin

	a_uns <= unsigned('0' & a);
	b_uns <= unsigned('0' & b);

	process (clk)
	begin
		if rising_edge(clk) then
			if a_uns > b_uns then 
				comp <= '1';
			else
				comp <= '0';
			end if;
			sum <= std_logic_vector(a_uns + b_uns);
		end if;
	end process;

end architecture rtl;
--------------------------------------------------------------



Example 10.3. Carry-ripple adder built with full-adder components

	--------------------------------------------------------------
library ieee;
use ieee.std_logic_1164.all;

entity full_adder_unit is	
	port (
		in1, in2, cin: in std_logic;
		sum, cout: out std_logic);
end entity;

architecture boolean of full_adder_unit is
begin
	sum <= in1 xor in2 xor cin;
	cout <= (in1 and in2) or (in1 and cin) or (in2 and cin);
end architecture;
--------------------------------------------------------------

--------------------------------------------------------------
library ieee;
use ieee.std_logic_1164.all;

entity carry_ripple_adder is	
	generic (
		NUM_BITS: natural := 8);
	port (
		a, b: in std_logic_vector(NUM_BITS-1 downto 0);
		cin: in std_logic;
		sum: out std_logic_vector(NUM_BITS-1 downto 0);
		cout: out std_logic);
end entity;

architecture structural of carry_ripple_adder is
	signal carry: std_logic_vector(0 to NUM_BITS);
begin
	carry(0) <= cin;
	gen_adder: for i in 0 to NUM_BITS-1 generate
		adder: entity work.full_adder_unit 
			port map (a(i), b(i), carry(i), sum(i), carry(i+1));
	end generate;
	cout <= carry(NUM_BITS);
end architecture;
--------------------------------------------------------------



Example 10.4. Hamming-weight calculator

	----------------------------------------------------------------------------------
library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity hamming_weight_calculator is
	generic (
		BITS_IN: positive := 16;
		BITS_OUT: positive := 5); --calculated by user as ceil(log2(BITS_IN+1))
	port (
		inp_vector: in std_logic_vector(BITS_IN-1 downto 0);
		hamm_weight: out std_logic_vector(BITS_OUT-1 downto 0));
end entity;

architecture concurrent of hamming_weight_calculator is
	type integer_array is array (0 to BITS_IN) of integer range 0 to BITS_IN; 
	signal internal: integer_array;
begin
	internal(0) <= 0;
	gen: for i in 1 to BITS_IN generate
		internal(i) <= internal(i-1) + 1 when inp_vector(i-1) else internal(i-1);
	end generate;
	hamm_weight <= std_logic_vector(to_unsigned(internal(BITS_IN), BITS_OUT));
end architecture;
----------------------------------------------------------------------------------



	use ieee.math_real.all;
...
	generic (
		BITS_IN: positive := 16;
		BITS_OUT: positive := integer(ceil(log2(real(BITS_IN+1))))); --a dependent constant
	port (...



	entity hamming_weight_calculator is
	generic (
		BITS_IN: positive := 16);
	port (
		inp_vector: in std_logic_vector(BITS_IN-1 downto 0);
		hamm_weight: out std_logic_vector(integer(ceil(log2(real(BITS_IN+1))))-1 downto 0)));
end entity;

architecture concurrent of hamming_weight_calculator is
	constant BITS_OUT: positive := integer(ceil(log2(real(BITS_IN+1))));
	...



Example 10.5. Signed integer adder

	----------------------------------------------------------------
library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity adder_signed is
	generic (
		NUM_BITS: integer := 4); 
	port (
		a, b: in std_logic_vector(NUM_BITS-1 downto 0);
		cin: in std_logic;
		sum: out std_logic_vector(NUM_BITS-1 downto 0);
		sumMSB, cout, oflow: out std_logic);
end entity;

architecture suggested of adder_signed is
	signal sum_sig: signed(NUM_BITS downto 0);
begin

	--Sign-extension, conversion to signed, and addition:
	sum_sig <= signed(a(NUM_BITS-1) & a) + signed(b) + cin;
	--sum_sig <= resize(signed(a), NUM_BITS+1) + signed(b) + cin; 
	
	--Conversion to std_logic_vector and final operations:
	sum <= std_logic_vector(sum_sig(NUM_BITS-1 downto 0));
	sumMSB <= sum_sig(NUM_BITS);
	cout <= a(NUM_BITS-1) xor b(NUM_BITS-1) xor sumMSB;
	oflow <= sum_sig(NUM_BITS) xor sum_sig(NUM_BITS-1);

end architecture;
----------------------------------------------------------------



Example 11.2. Programmable combinational delay line (structural)

	--The component (delay block)
----------------------------------------------------------------
library ieee;
use ieee.std_logic_1164.all;

entity delay_block is
	generic (
		BLOCK_LENGTH: natural);
	port (
		din: in std_logic;
		sel: in std_logic;
		dout: out std_logic); 
end entity;

architecture delay_block of delay_block is 
	signal node_vector: std_logic_vector(0 to 2*BLOCK_LENGTH);
	attribute keep: boolean;
	attribute keep of node_vector: signal is true;
begin
	node_vector(0) <= din;
	gen: for i in 1 to 2*BLOCK_LENGTH generate
		node_vector(i) <= not node_vector(i-1);
	end generate;
	dout <= node_vector(2*BLOCK_LENGTH) when sel else din;
end architecture;
----------------------------------------------------------------


--Main code (complete delay line)
----------------------------------------------------------------
library ieee;
use ieee.std_logic_1164.all;

entity delay_line is 
	generic (
		NUM_BLOCKS: natural := 4); --this is M in figure 11.2b
	port (
		din: in std_logic;
		--sel: in std_logic_vector(0 to NUM_BLOCKS-1);
		sel: in std_logic_vector(NUM_BLOCKS-1 downto 0);
		dout: out std_logic); 
end entity;

architecture structural of delay_line is 
	signal node_vector: std_logic_vector(0 to NUM_BLOCKS);
	attribute keep: boolean;
	attribute keep of node_vector: signal is true;
begin
	node_vector(0) <= din;
	gen: for i in 1 to NUM_BLOCKS generate
		blocki: entity work.delay_block 
			generic map (2**(i-1)) 
			port map (node_vector(i-1), sel(i-1), node_vector(i));
	end generate;
	dout <= node_vector(NUM_BLOCKS);
end architecture;
----------------------------------------------------------------



Example 12.3. Counter with is_max flag and RTL analysis

	----------------------------------------------------------------
library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity counter_with_is_max_flag is
	generic (
		MAX: natural := 6;
		BITS: natural := 3); --not independent (see section 6.7)
	port (
		clk: in std_logic;
		count: out std_logic_vector(BITS-1 downto 0);
		is_max: out std_logic);
end entity;	

architecture rtl of counter_with_is_max_flag is
	signal i: natural range 0 to MAX;
begin

	P1: process (clk)
	begin
		if rising_edge(clk) then
			if i=MAX-1 then
				is_max <= '1';
				i <= i + 1;
			elsif i=MAX then
				is_max <= '0';
				i <= 0;
			else
				i <= i + 1;
			end if;
		end if;
	end process;
	count <= std_logic_vector(to_unsigned(i, BITS));

end architecture;
----------------------------------------------------------------

		P2: process (clk)
	begin
		if rising_edge(clk) then
	
			--Counter:
			if i /= MAX then
				i <=  i + 1;
			else
				i <= 0;
			end if;

			--Flag generator:
			if i=MAX-1 then
				is_max <= '1';
			else
				is_max <= '0';
			end if;

		end if;
	end process;



Example 12.4. Counters with signal and variable

	--------------------------------------------------
entity dual_counter is
	port (
		clk: in bit;
		count1, count2: out natural range 0 to 10);
end entity;

architecture rtl of dual_counter is
begin

	counter1: process (clk)
	begin
		if clk'event and clk='1' then
			count1 <= count1 + 1;
			if count1=10 then 
				count1 <= 0;
			end if;
		end if;
	end process counter1;

	counter2: process (clk)
		variable var: natural range 0 to 10;
	begin
		if clk'event and clk='1' then
			var := var + 1;
			if var=10 then 
				var := 0;
			end if;
		end if;
		count2 <= var;
	end process counter2;

end architecture;
--------------------------------------------------



Example 12.10. Recommended shift register implementation

	----------------------------------------------------------------------------------------
library ieee;
use ieee.std_logic_1164.all;

entity shift_register is
	generic (
		NUM_BITS: natural := 8;   
		NUM_STAGES: natural := 4);
	port (
		clk: in std_logic;
		din: in std_logic_vector(NUM_BITS-1 downto 0);
		dout: out std_logic_vector(NUM_BITS-1 downto 0));
end entity;	
  
architecture recommended of shift_register is
	type slv_array is array (0 to NUM_STAGES-1) of std_logic_vector(NUM_BITS-1 downto 0);
	signal q: slv_array;
begin

	process (clk)
	begin
		if rising_edge(clk) then
			q <= din & q(0 to NUM_STAGES-2);
		end if;
		dout <= q(NUM_STAGES-1);
	end process;

end architecture;
----------------------------------------------------------------------------------------



Example 13.1. Generic tree-type adder array

	-------------------------------------------------------------
library ieee;
use ieee.numeric_std.all;
package user_defined_type_pkg is
	type signed_vector is array (natural range <>) of signed;
end package;
-------------------------------------------------------------

----------------------------------------------------------------------------------------
library ieee;
use ieee.numeric_std.all;
use ieee.math_real.all;
use work.user_defined_type_pkg.all;

entity adder_array_generic_tree is
	generic (
		NUM_INPUTS: natural := 10; 
		NUM_BITS: natural := 7);
	port (
		x: in signed_vector(0 to NUM_INPUTS-1)(NUM_BITS-1 downto 0);
		sum: out signed(NUM_BITS+integer(ceil(log2(real(NUM_INPUTS))))-1 downto 0));
   end entity;

architecture tree_type_generic of adder_array_generic_tree is
	constant LAYERS: natural := integer(ceil(log2(real(NUM_INPUTS))));
	constant PWR_OF_TWO: natural := 2**LAYERS;
	alias EXTRA_BITS: natural is LAYERS;
begin

	process (all)
		variable accum: signed_vector(0 to PWR_OF_TWO-1)(NUM_BITS+EXTRA_BITS-1 downto 0);
	begin
		
		--Initialization:
		loop1: for i in 0 to NUM_INPUTS-1 loop
			accum(i) := resize(x(i), NUM_BITS+EXTRA_BITS);
		end loop loop1;
		accum(NUM_INPUTS to PWR_OF_TWO-1) := (others => (others => '0'));
	
		--Generic tree-type adder array:
		loop3: for j in 1 to LAYERS loop
			loop4: for i in 0 to PWR_OF_TWO/(2**j)-1 loop
				accum(i) := accum(2*i) + accum(2*i+1);
			end loop loop4;	
      end loop loop3;
		sum <= accum(0);

	end process;
	
end architecture;
----------------------------------------------------------------------------------------



Example 13.2. Single-switch debouncer

	----------------------------------------------------------------------------------------
library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;
use ieee.math_real.all;

entity debouncer is
	generic (
		T_DEB_MS: natural := 25;        --minimum debounce time in ms
		F_CLK_KHZ: natural := 50_000);  --clock frequency in kHz
	port (
		x, clk: in std_logic;
		y: out std_logic);
end entity;

architecture single_switch of debouncer is 
	constant COUNTER_BITS: natural := 1 + integer(ceil(log2(real(T_DEB_MS*F_CLK_KHZ))));
	signal x_reg: std_logic;
begin

	process (clk)
		variable count: unsigned(COUNTER_BITS-1 downto 0);
   begin

		--Timer (with input register):
		if rising_edge(clk) then
			x_reg <= x;
			if y=x_reg then
				count := (others => '0');
			else
				count := count + 1;
			end if;
		end if;

		--Output register:
		if falling_edge(clk) then
			if count(COUNTER_BITS-1) then
				y <= not y;
			end if;
		end if;

	end process;

end architecture;
----------------------------------------------------------------------------------------



Example 14.2. Function ceil_log2 in a package

	----------------------------------------------------------
package subprograms_pkg is	
	function ceil_log2 (input: positive) return natural;
end package;

package body subprograms_pkg is
	function ceil_log2 (input: positive) return natural is
		variable result: natural := 0;
	begin
		while 2**result < input loop
			result := result + 1;
		end loop;
		return result;
	end function ceil_log2;
end package body;
----------------------------------------------------------


----------------------------------------------------------
use work.subprograms_pkg.all;

entity test_circuit is
	generic (
		BITS: natural := 8);	
	port (
		inp: in positive range 1 to 2**BITS-1;
		outp: out natural range 0 to BITS);
end entity;

architecture test_circuit of test_circuit is
begin
	outp <= ceil_log2(inp);
end architecture;
----------------------------------------------------------



Example 14.4. Function slv_to_integer in a process

library ieee;
use ieee.std_logic_1164.all;

entity test_circuit is
	generic (
		WIDTH: natural := 8);	
	port (
		clk: in std_logic;
		inp: in std_logic_vector(WIDTH-1 downto 0);
		outp: out integer range 0 to 2**WIDTH-1);
end entity;

architecture test_circuit of test_circuit is
begin

	process
		function slv_to_integer (slv: std_logic_vector) return integer is
			variable result: integer range 0 to 2**slv'length-1 := 0;	
		begin
			for i in slv'range loop
				result := result*2;
				if slv(i)='1' or slv(i)='H' then 
					result := result + 1;
				end if;
			end loop;
			return result;
		end function slv_to_integer;
	begin
		wait until clk;
		outp <= slv_to_integer(inp);
	end process;

end architecture;
---------------------------------------------------------------------------



Example 16.2. Garage door controller

library ieee;
use ieee.std_logic_1164.all;

entity garage_door_controller is
	port (
		clk, rst: in std_logic;
		remote, door_closed, door_open: in std_logic;
		motor, direction: out std_logic);
end entity;

architecture fsm of garage_door_controller is 
	type state_type is (ready_to_open, opening1, opening2, 
		opening3, ready_to_close, closing1, closing2, closing3); 
	signal state: state_type;
begin

	--Logic + register for state:
	process (clk, rst)
	begin
		if rst then
			state <= ready_to_open;
		elsif rising_edge(clk) then
			case state is
				when ready_to_open =>
					if door_open then
						state <= ready_to_close;
					elsif remote then
						state <= opening1;
					else
						state <= ready_to_open;
					end if;
				when opening1 =>
					if not remote then
						state <= opening2;
					else
						state <= opening1;
					end if;			
				when opening2 =>
					if door_open then
						state <= ready_to_close;
					elsif remote then
						state <= opening3;
					else
						state <= opening2;
					end if;	
				when opening3 =>
					...	
				when ready_to_close =>
					...	
				when closing1 =>
					...			
				when closing2 =>
 					...		
				when closing3 =>
					...		
			end case;
		end if;
	end process;

	--Logic + register for outputs:
	process (clk)
	begin
		if rising_edge(clk) then
			case state is
				when ready_to_open =>
					motor <= '0';
					direction <= '-';
				when opening1 =>
					motor <= '1';
					direction <= '1';
				when opening2 =>
					motor <= '1';
					direction <= '1';
				when opening3 =>
					motor <= '0';
					direction <= '-';
				when ready_to_close =>
					motor <= '0';
					direction <= '-';
				when closing1 =>
					motor <= '1';
					direction <= '0';
				when closing2 =>
					motor <= '1';
					direction <= '0';
				when closing3 =>
					motor <= '0';
					direction <= '-';
			end case;
		end if;
	end process;

	--Notice that default values could have been used in the process above
		
end architecture;
---------------------------------------------------------------------------



Example 16.4. SPI interface for an A/D converter

	--------------------------------------------------------------------------
library ieee;
use ieee.std_logic_1164.all;

entity SPI_interface_for_MAX1118 is
	port (
		clk, rst: in std_logic;
		rd: in std_logic;
		SSn, SCLK: out std_logic;
		MISO: in std_logic;
		received_data: out std_logic_vector(7 downto 0));
end entity;

architecture fsm of SPI_interface_for_MAX1118 is
	signal clk_5MHz: std_logic;
	type state_type is (idle, convert, read_data, hold);
	signal pr_state, nx_state: state_type;
	signal t, tmax: natural range 0 to 37;
begin

	--Generate SPI clock (5 MHz):
	process (clk)
		variable count: natural range 0 to 4;
	begin
		if rising_edge(clk) then
			if count=4 then
				clk_5MHz <= not clk_5MHz;
				count := 0;
			else
				count := count + 1;
			end if;
		end if;
	end process;
	
	--Register for machine state:
	process (clk_5MHz, rst)
	begin
		if rst then
			pr_state <= idle;
		elsif rising_edge(clk_5MHz) then
			pr_state <= nx_state;
		end if;
	end process;

	--Logic for machine state:
	process (all)
	begin
		case pr_state is
 			when idle =>
				if rd then
					nx_state <= convert;
				else
					nx_state <= idle;
				end if;
			when convert =>
				if t=tmax then
					nx_state <= read_data;
				else
					nx_state <= convert;
				end if;
			when read_data =>
				if t=tmax then
					nx_state <= hold;
				else
					nx_state <= read_data;
				end if;
			when hold =>
				if not rd then
					nx_state <= idle;
				else
					nx_state <= hold;
				end if; 
		end case;
	end process;

	--Logic for machine outputs:
	process (all)
	begin
		case pr_state is
			when idle => 
				SSn <= '1';
				SCLK <= '1';
			when convert => 
				SSn <= '0';
				SCLK <= '1';
			when read_data => 
				SSn <= '0';
				SCLK <= clk_5MHz;
			when hold => 
				SSn <= '1';
				SCLK <= '1';
		end case;
	end process;	

	--Store data read from ADC:
	process (clk_5MHz)
	begin
		if rising_edge(clk_5MHz) then
			if pr_state = read_data then
				received_data(7-t) <= MISO;
			end if;
		end if;
	end process;

	--Timer:
	process (all)
	begin
		case pr_state is
			when convert => tmax <= 37;
			when read_data => tmax <= 7;
			when others => tmax <= 0;
		end case;
		if rising_edge(clk_5MHz) then
			if pr_state /= nx_state then
				t <= 0;
			elsif t /= tmax then
				t <= t + 1;
			end if;
		end if;
	end process;
	
end architecture;
--------------------------------------------------------------------------



Example 17.1. SPI Interface for an EEPROM Device (with FSM)

	-----------------------------------------------------------------------------
library ieee;
use ieee.std_logic_1164.all;

entity SPI_interface_for_EEPROM is
	generic (
		WREN_OPCODE: std_logic_vector(7 downto 0) := "00000110";
		WRITE_OPCODE: std_logic_vector(7 downto 0) := "00000010";
		READ_OPCODE: std_logic_vector(7 downto 0) := "00000011";
		INITIAL_WRITE_ADDR: std_logic_vector(15 downto 0) := (others => '0');
		INITIAL_READ_ADDR: std_logic_vector(15 downto 0) := (others => '0'));
	port (
		clk, rst: in std_logic;
		wr, rd: in std_logic;
		SSn: out std_logic;
		SCLK: out std_logic;
		MOSI: out std_logic;
		MISO: in std_logic;
		data_read_from_EEPROM: out std_logic_vector(7 downto 0));
end entity;

architecture fsm of SPI_interface_for_EEPROM is

	--Reference clock (12.5 MHz):
	signal clk_12MHz: std_logic; 
	
	--FSM-related declarations:	
	type state_type is (idle, WREN_OP, deselect1, deselect2, WRITE_OP, 
		initial_wr_addr, 	write_data, wait_wr_low, READ_OP, initial_rd_addr,
		read_data, wait_rd_low); 
	signal pr_state, nx_state: state_type;
	
	--Timer-related declarations:
	signal i, imax: natural range 0 to 15;
	signal j, jmax: natural range 0 to 4;
	
	--Test data (to be written to and read from EEPROM):
	type data_array is array (natural range <>) of std_logic_vector;
	constant test_data_out: data_array(0 to 4)(7 downto 0) :=
		("00000001", "10000000", "00111100", "11000011", "11111111");
	signal test_data_in: data_array(0 to 4)(7 downto 0);
	
begin

	--Generate SPI clock (12.5MHz) from system clock (50MHz):
	process (clk)
		variable count: natural range 0 to 1;
	begin
		if rising_edge(clk) then
			if count=1 then
				clk_12MHz <= not clk_12MHz;
				count := 0;
			else
				count := count + 1;
			end if;
		end if;
	end process;
	
	--Dual timer (i=primary, j=secondary):
	process (all)
	begin
		--Define imax and jmax values:
		case pr_state is
			when WREN_OP | WRITE_OP | write_data | READ_OP | read_data => imax <= 7;
			when initial_wr_addr | initial_rd_addr => imax <= 15;
			when others => imax <= 0;
		end case;
		case pr_state is
			when write_data | read_data => jmax <= 4;
			when others => jmax <= 0;
		end case;
		--Implement pointers i and j:
		if rst then
			i <= 0;
			j <= 0;
		elsif falling_edge(clk_12MHz) then
			if pr_state /= nx_state then
				i <= 0;
				j <= 0;
			elsif not (i=imax and j=jmax) then
				if i/=imax then
					i <= i + 1;
				elsif j/=jmax then
					i <= 0;
					j <= j + 1;
				end if;
			end if;
		end if;
	end process;
	
	--State register (block R1 of figure 15.2b):
	process (clk_12MHz, rst)
	begin
		if rst then 
			pr_state <= idle;
		elsif falling_edge(clk_12MHz) then
			pr_state <= nx_state;
		end if;
	end process;
	
	--State logic (block L1 of figure 15.2b):
	process (all)
	begin
		case pr_state is
			when idle =>
				if wr and not rd then 
					nx_state <= WREN_OP;
				elsif rd and not wr then 
					nx_state <= READ_OP;				
				else 
					nx_state <= idle;
				end if;
			when WREN_OP =>
				if i=imax then
					nx_state <= deselect1;
				else
					nx_state <= WREN_OP;
				end if;
			when deselect1 =>
				nx_state <= deselect2;
			when deselect2 =>
				nx_state <= WRITE_OP;
			when WRITE_OP =>
				if i=imax then
					nx_state <= initial_wr_addr;
				else
					nx_state <= WRITE_OP;
				end if;
			when initial_wr_addr =>
				...								
			when write_data =>
				...					
			when wait_wr_low =>
				...					
			when READ_OP =>
				...				
			when initial_rd_addr =>
				...
			when read_data =>
				...				
			when wait_rd_low =>
				...				
		end case;
	end process;

	--Output logic (block L2' of figure 15.2b):
	process (all)
	begin
		--Default values:
		SSn <= '0';
		SCLK <= clk_12MHz;
		MOSI <= '-';
		--Other values:
		case pr_state is 
			when idle =>
				SSn <= '1';
				SCLK <= '0';
			when WREN_OP =>
				MOSI <= WREN_OPCODE(7-i);
			when deselect1 =>
				SCLK <= '0';	
			when deselect2 =>
				SSn <= '1';
				SCLK <= '0';
			when WRITE_OP =>
				MOSI <= WRITE_OPCODE(7-i);
			when initial_wr_addr =>
				MOSI <= INITIAL_WRITE_ADDR(15-i);
			when write_data =>
				MOSI <= test_data_out(j)(7-i);
			when wait_wr_low =>
				SCLK <= '0';	
			when READ_OP =>
				MOSI <= READ_OPCODE(7-i);
			when initial_rd_addr =>
				MOSI <= INITIAL_READ_ADDR(15-i);			
			when wait_rd_low =>
				SCLK <= '0';
		end case;
	end process;
	
	--Store data read from EEPROM:
	process (clk_12MHz)
	begin		
		if rising_edge(clk_12MHz) then	
			if pr_state=read_data then
				test_data_in(j)(7-i) <= MISO;
			end if;
		end if;
	end process; 
	
	--Display data read from EEPROM (test circuit @1Hz):
	process (clk_12MHz)
		variable count1: natural range 0 to 12_500_000;
		variable count2: natural range 0 to 4;
	begin
		if falling_edge(clk_12MHz) then
			if pr_state=idle then
				if count1=12_500_000 then 
					count1 := 0;
					if count2 /= 4 then
						count2 := count2 + 1;
					else	
						count2 := 0;
					end if;
				else
					count1 := count1 + 1;
				end if;
				data_read_from_EEPROM <= test_data_in(count2);
			end if;
		end if;
	end process;
	
end architecture;
-----------------------------------------------------------------------------



Example 17.3. I2C Interface for an A/D Converter (with Pointer)

library ieee;
use ieee.std_logic_1164.all;

entity I2C_interface is
	generic (
		SLAVE_ADDRESS: std_logic_vector(6 downto 0) := "0110110";
		SETUP_REGISTER: std_logic_vector(7 downto 0) := "1101001-";
		CONFIG_REGISTER: std_logic_vector(7 downto 0) := "011---01");
	port (
		clk, rst: in std_logic;
		wr, rd: in std_logic;
		SCL: out std_logic;
		SDA: inout std_logic;
		received_data: out std_logic_vector(9 downto 0));
end entity;

architecture fsm of I2C_interface is
	signal clk_400kHz: std_logic;			--reference clock
	signal i: natural range 0 to 34; 	--pointer
	signal wr_enable, rd_enable: std_logic;
begin

	--Generate 400kHz from 50MHz system clock:
	process (clk)
		variable count: natural range 0 to 62;
	begin
		if rising_edge(clk) then
			if count=62 then
				clk_400kHz <= not clk_400kHz;
				count := 0;
			else
				count := count + 1;
			end if;
		end if;
	end process;
	
	--Generate pointer and enable signals:
	process (clk_400kHz, rst)
	begin
		if rst then
			wr_enable <= '0';
			rd_enable <= '0';
			i <= 0;
		elsif falling_edge(clk_400kHz) then
			if (wr and not rd_enable) or wr_enable then
				if i<30 then
					wr_enable <= '1';
					i <= i + 1;
				elsif not wr then
					wr_enable <= '0';
					i <= 0;
				end if;
			elsif (rd and not wr_enable) or rd_enable then
				if i<34 then
					rd_enable <= '1';
					i <= i + 1;
				elsif not rd then
					rd_enable <= '0';
					i <= 0;
				end if;
			else
				wr_enable <= '0';
				rd_enable <= '0';
				i <= 0;
			end if;
		end if;
	end process;
	
	--Generate SCL and SDA signals:
	process (all)
	begin
		
		if wr_enable then
			--Define SCL for writing:
			case i is
				when 1 | 30 => SCL <= '1';
				when others => SCL <= clk_400kHz;
			end case;
			--Define SDA for writing:
			case i is
				--Start:
				when 1 => SDA <= '0';
				--Slave address to write:
				when 2 to 8 => SDA <= SLAVE_ADDRESS(8-i);
				when 9 => SDA <= '0';
				when 10 => SDA <= 'Z';
				--Setup register:
				when 11 to 18 => SDA <= SETUP_REGISTER(18-i);
				when 19 => SDA <= 'Z';
				--Configuration register:
				when 20 to 27 => SDA <= CONFIG_REGISTER(27-i);
				when 28 => SDA <= 'Z';
				--Stop:
				when 29 => SDA <= '0';		
				when others => SDA <= '1';			
			end case;
		
		elsif rd_enable then
			--Define SCL for reading:
			case i is
				when 1 | 34 => SCL <= '1';
				when 11 to 14 => SCL <= '0';
				when others => SCL <= clk_400kHz;
			end case;
			--Define SDA for reading:
			case i is
				--Start:
				when 1 => SDA <= '0';
				--Slave address to read:
				when 2 to 8 => SDA <= SLAVE_ADDRESS(8-i);
				when 9 => SDA <= '1';
				when 10 => SDA <= 'Z';
				--Clock stretch:
				when 11 to 14 => SDA <= 'Z';
				--Read result byte 1:
				when 15 to 22 => SDA <= 'Z';		
				when 23 => SDA <= '0';
				--Read result byte 0:	
				when 24 to 31 => SDA <= 'Z';
				when 32 => SDA <= '1';
				--Stop:
				when 33 => SDA <= '0';
				when others => SDA <= '1';				
			end case;
			
		else
			SCL <= '1';
			SDA <= '1';
		end if;

	end process;
	
	--Store data read from ADC:
	process (clk_400kHz)
	begin		
		if rising_edge(clk_400kHz) then	
			if rd_enable then
				if i=21 then
					received_data(9) <= SDA;
				elsif i=22 then
					received_data(8) <= SDA;
				elsif i>23 and i<32 then
					received_data(31-i) <= SDA;
				end if;
			end if;
		end if;
	end process; 
	
end architecture;
---------------------------------------------------------------------------



Example 17.6. VGA Video Interface for a Hardware-Generated Image 

library ieee;
use ieee.std_logic_1164.all;

entity image_generator_plus_vga_interface is	
	generic (
		H_LOW: natural := 96;
		HBP: natural := 48;
		H_HIGH: natural := 640;
		HFP: natural := 16;
		V_LOW: natural := 2;
		VBP: natural := 33;
		V_HIGH: natural := 480;
		VFP: natural := 10);
	port (
		clk: in std_logic; 			--50MHz system clock
		clk_vga: out std_logic;	--25MHz pixel clock 
		R_switch, G_switch, B_switch: in std_logic;
		Hsync, Vsync: out std_logic;
		R, G, B: out std_logic_vector(9 downto 0);
		BLANKn, SYNCn : out std_logic);
end entity;

architecture rtl of image_generator_plus_vga_interface is
	signal Hactive, Vactive, dena: std_logic;
begin

	--CIRCUIT 1: CONTROL GENERATOR

	--Static signals for DAC:
	BLANKn <= '1';  --no blanking
	SYNCn <= '0';   --no sync on green

	--Create VGA clock (50MHz -> 25MHz):
	process (clk)
	begin
		if rising_edge(clk) then
			clk_vga <= not clk_vga;
		end if;
	end process;

	--Create horizontal signals:
	process (clk_vga)
		variable Hcount: natural range 0 to H_LOW + HBP + H_HIGH + HFP;
	begin
		if rising_edge(clk_vga) then
			Hcount := Hcount + 1;
			if Hcount = H_LOW then
				Hsync <= '1';
			elsif Hcount = H_LOW + HBP then
				Hactive <= '1';
			elsif Hcount = H_LOW + HBP + H_HIGH then
				Hactive <= '0';
			elsif Hcount = H_LOW + HBP + H_HIGH + HFP then
				Hsync <= '0';
				Hcount := 0;
			end if;
		end if;
	end process;

	--Create vertical signals:
	process (Hsync)
		variable Vcount: natural range 0 to V_LOW + VBP + V_HIGH + VFP;
	begin
		if rising_edge(Hsync) then
			Vcount := Vcount + 1;
			if Vcount = V_LOW then
				Vsync <= '1';
			elsif Vcount = V_LOW + VBP then
				Vactive <= '1';
			elsif Vcount = V_LOW + VBP + V_HIGH then
				Vactive <= '0';
			elsif Vcount = V_LOW + VBP + V_HIGH + VFP then
				Vsync <= '0';
				Vcount := 0;
			end if;
		end if;
	end process;

	--Enable diplay:
	dena <= Hactive and Vactive;

	--CIRCUIT 2: IMAGE GENERATOR

	process (all)
		variable line_count: natural range 0 to V_HIGH;
	begin
		if rising_edge(Hsync) then
			if Vactive then
				line_count := line_count + 1;
			else
				line_count := 0;
			end if;
		end if;
		if dena then
			case line_count is
				when  0 =>
					R <= (others => '1');
					G <= (others => '0');
					B <= (others => '0');
				when 1 | 2 | 479 =>
					R <= (others => '0');
					G <= (others => '1');
					B <= (others => '0');
				when 3 to 5 =>
					R <= (others => '0');
					G <= (others => '0');
					B <= (others => '1');
				when others =>
					R <= (others => R_switch);
					G <= (others => G_switch);
					B <= (others => B_switch);
			end case;
		else
			R <= (others => '0');
			G <= (others => '0');
			B <= (others => '0');		
		end if;
	end process;

end architecture;
------------------------------------------------------------

The following editable VHDL codes are included next:
Example 6.3. Registered add-and-compare circuit

Example 10.3. Carry-ripple adder built with full-adder components
Example 10.4. Hamming-weight calculator

Example 10.5. Signed integer adder

Example 11.2. Programmable combinational delay line (structural)
Example 12.3. Counter with is_moax flag and RTL analysis

Example 12.4. Counters with signal and variable

Example 12.10. Recommended shift register implementation
Example 13.1. Generic tree-type adder array

Example 13.2. Single-switch debouncer

Example 14.2. Function ceil_log2 in a package

Example 14.4. Function slv_to_integer in a process

Example 16.2. Garage door controller

Example 16.4. SP! interface for an A/D converter

Example 17.1. SPI Interface for an EEPROM Device (with FSM)
Example 17.3. IC Interface for an A/D Converter (with Pointer)
Example 17.6. VGA Video Interface for a Hardware-Generated Image




